

**Doc No.:** AN-225

**Version:** 1.1

**Date:** 08 July 2013

**Subject:** Move Spiral

## APPLICATION NOTE

[www.triomotion.com](http://www.triomotion.com)

### 1. Introduction

Some applications require spiral motion for example in glue laying, some customers have asked how this can be achieved using a TRIO Coordinator without having to program the movement into a CAM profile.

This application note describes how a spiral profile can be created on an XY machine using 2 virtual axes.

### 2. Spiral specifying




Figure 1 Spiral Specification

This program generates a fixed pitch spiral as per Figure 1. The spiral path speed is kept constant throughout the move.

There are 3 parameters to specify the spiral profile:

- Diameter:            The outside diameter of the spiral
- No\_turns:           The number of revolutions
- Thickness:           the thickness of the layer

The program will then calculate the inside diameter and display an error if the inside diameter is less than 0. so in Figure 1 the outside diameter could be 100, no\_turns is 4 and the thickness could be 10.

### 3. Spiral Program

The spiral is generated by moving 2 virtual axis through the no\_turns of MOVECIRC. By varying the connect ratio the real axis perform a spiral. The initial speed on the virtual axis specifies the path speed of the spiral. The program performs a loop that decreases the connect ratio, and increases the virtual axis speed. This keeps the path speed constant on the real axis. To keep the spiral constant thickness the loop cycle time is reduced proportionally to the connect ratio.

```

=====
' Customer: Application Example
' Application: MOVESPIRAL
' Module: SPIRAL.BAS
' Platform: Trio Motion Coordinator
-----
' Version: 0.1 Issued: 11th February 2008
' Author: Alex Hazle
' Original: 11th February 2008
-----
' Copyright (c) 1999-2008 Trio Motion Technology ltd.
,
' Trio Motion Technology Ltd.
' Web site: www.triomotion.com
=====
' Revision History:
' V0.1 1st release
=====
' This example code can be used within a program to generate a spiral movement
' on an XY machine. The Spiral will move with constant thickness AND constant
' path speed. You will need 2 virtual axis to complete this.
=====

xaxis=0 'real x axis
yaxis=1 'real y axis
ximg=2 'virtual x AXIS
yimg=3 'virtual y axis
'Connect the axis
BASE(xaxis)
CONNECT(1,ximg)
BASE(yaxis)
CONNECT(1,yimg)
BASE(ximg,yimg)
MERGE=ON 'enables smooth transition of circles
diameter=120 'outer diameter of spiral
SPEED=100 'path speed in units of spiral
ACCEL = SPEED * 10
DECEL = ACCEL
no_turns=5 'no of turns of spiral
thickness=10 'thickness between spiral turns
'Check the inner diameter of the spiral is not less than zero
inner_diamter=diameter-2*no_turns*thickness
PRINT#6, "Inner Diameter=";inner_diamter
IF inner_diamter<0 THEN
PRINT#6, "INNER DIAMETER LESS THAN ZERO"
GOTO error

```

```

ENDIF
'calculate the length of the circle moves for time parameter
length=PI*diameter*no_turns
PRINT#6, "Length= ";length
time_val=length/SPEED
'calculate the increment change of the connection ratio
change=(1-(inner_diameter/diameter))/(time_val/0.1)
PRINT#6, "Change increment= ";change
turns_done=1 'turns counter
x=1 'connection variable
MOVECIRC(0,0,(diameter/2),0,1)
WAIT UNTIL MTYPE AXIS(2)<>0
WHILE MTYPE AXIS(2)<>0
    x=x-change 'update connection ratio
    BASE(ximg,yimg)
    IF x>=0 THEN
'adjust imaginary axis speed to maintain constant spiral path SPEED
        SPEED=(1/x)*length/time_val
    ENDIF
    IF NTYPE=0 AND turns_done < no_turns THEN
        MOVECIRC(0,0,(diameter/2),0,1)' add new movecirc TO NTYPE IF required
        turns_done=turns_done+1
    ENDIF
'update the connect ratio
    BASE(xaxis)
    CONNECT(x,ximg)
    BASE(yaxis)
    CONNECT(x,yimg)
'modify loop time to maintain constant thickness spiral
    WA(100*x)
WEND
'rapidstop to remove connects, you could use cancel
RAPIDSTOP(2)
error: 'exit position if inner diameter is less than 0

```

## 4. 3D Spirals

The above program can be easily modified to perform 3D spirals with constant path speed. Simply add a height parameter and change the MOVECIRC commands to MHELIX with the parameters as follows:

```
MHELIX(0,0,(diameter/2),0,1,(height/no_turns),1)
```

Where height is the z axis movement of the spiral and the final 1 specifies the type of helix move. Initialisation of Height should be added to the parameters section as well.

## 5. Limitations

As the program operates through a loop there is a finite resolution, this should only affect very large diameters with large numbers of turns. You should also remember that it is not a true spiral but many segments of a circle with ever decreasing diameters